

UNIVERSITY OF BIRMINGHAM



Department of Metallurgy & Materials

Activity Manual

Contents

1	Background	4
1.1	Ion Irradiation	4
1.1.1	Ion Irradiation at the University of Birmingham	4
1.1.2	Simulating Ion Irradiation with SRIM	4
1.1.3	Transmutation of Nuclei by Ion Irradiation	5
1.2	Decay and Activity Equations	6
1.2.1	Radioactive Decay	6
1.2.2	Bateman Equation for Radioactive Decay	6
1.2.3	Laplace Transform	7
1.2.4	Constructing the Differential Equations	7
1.2.5	Numerical Inversion of the Laplace Transform	8
1.2.6	Analytic Solution by Partial Fraction Expansion	9
1.2.7	Preference: Analytic over Numeric	11
1.3	Computational Methods	12
1.3.1	Activity Code	12
1.3.2	Approximations	13
1.3.3	Results	13
2	Installation and Using the Activity Code	17
2.1	Getting Started	17
2.1.1	Prerequisites	17
2.2	Installing Activity	17
2.2.1	Download Source Code	17
2.2.2	Compile Source Code	18
2.3	Input File	18
2.4	Acknowledgements	21
	Appendices	22

A Example Input File **23**

 A.1 Iron 36MeV Proton Beam 23

B Fortran 90 Code **25**

 B.1 Fortran 90 Implementation of Analytic Method 25

List of Figures

1.1	Proton energy loss in Fe simulated with SRIM [2]	5
1.2	An example decay chain from an unstable parent isotope, through unstable daughter isotopes ending with a stable daughter isotope.	6
1.3	An example of several decay chains including branching factors and possible external source terms for each isotope on each chain.	7
1.4	Decay of Po-218: Analytical and Geant4 Calculations [5]	11
1.5	Flow chart of major processes in the Activity code	12
1.6	Sample Activity code output chart for the top five most active isotopes for Iron irradiated by 36MeV protons.	14
1.7	Sample Activity code output chart for the expected gamma lines to be measured for Iron irradiated by 36MeV protons.	15
1.8	Sample Activity code output chart for the top five most active isotopes for Molybdenum irradiated by 13MeV protons.	15
1.9	Sample Activity code output chart for the expected gamma lines to be measured for Molybdenum irradiated by 13MeV protons.	16

Chapter 1

Background

1.1 Ion Irradiation

A computer program, Activity, was developed to predict the activity and gamma lines of materials irradiated with an ion beam. It uses the TENDL[1] proton reaction cross section database, the Stopping and Range of Ions in Matter (SRIM)[2] code, a Nuclear Data Services (NDS) radioactive decay database [3] and an ENDF gamma decay database [4]. An extended version of Bateman's equation is used to calculate the activity at time t , and this equation is solved analytically, with the option to also solve by numeric inverse Laplace transform as a failsafe. The program outputs the expected activity and gamma lines of the activated material.

1.1.1 Ion Irradiation at the University of Birmingham

The Scanditronix MC-40 Cyclotron is used at the University of Birmingham to create a beam of protons or other light ions. The energies of these ions are typically between 10 MeV and 60 MeV with beam currents ranging up to 50 microamps (3.1×10^{14} protons per second). Target materials are irradiated by this cyclotron for a number of reasons, including purposely creating radioactive isotopes for the nearby Queen Elizabeth Hospital, investigating ion irradiation damage and emulating neutron irradiation.

The Cyclotron is usually used to create radioactive isotopes for medical use, but an additional beam line has been devoted to material science investigations into radiation damage. While the creation of radioactive isotopes is desired in some cases, material being tested for radiation damage should preferably have low levels of radioactivity.

It is expensive to arrange the irradiation of target materials by high energy neutrons sources, whereas it is relatively inexpensive to irradiate using an ion beam on the MC-40 Cyclotron. The energies can be controlled, and a set dose at a single energy, or a range of energies, can be precisely deposited in the target material.

The Activity code discussed here was developed to calculate the activity of a target material irradiated by a proton beam. It has been developed in Fortran and uses data from the TENDL-2013 proton cross section database, SRIM ion transport code and NDS radioactive decay database.

1.1.2 Simulating Ion Irradiation with SRIM

A package of ion transport codes, SRIM, is freely available to download and use to investigate the transport of ions through matter. SRIM uses the binary collision approximation (BCA) to simulate the passage of ions in a material. It is an approximate method, and one key restriction is that it does not take into account the structure of the material, and this approximation is therefore also imposed on the Activity code.




Figure 1.1: Proton energy loss in Fe simulated with SRIM [2]

One file that SRIM creates is of importance to the Activity code, and that is the trajectory file that contains the energy and x,y,z co-ordinate data points for simulated ions moving through matter. Figure 1.1 shows the trajectory of protons passing through an Iron target, and it is this set of data points (together with the cross section database) that the Activity code uses to calculate the reaction rates for the transmutation of nuclei in the target. At higher energies, the ions slow as they lose energy due to electronic stopping, but as the ion energy drops the mechanism of loss through nuclear collisions becomes important. The spreading of ion depths at lower energies is a result of the higher momentum transfer during nuclear collisions, as can be seen in Figure 1.1.

1.1.3 Transmutation of Nuclei by Ion Irradiation

Considering a simplified nuclear potential well, energetic protons approaching a nucleus may overcome the coulomb potential barrier. They are captured by the nucleus and held within the potential well by the strong nuclear force. This process may leave the nucleus in an excited and unstable state, depending on the input energy of the proton and configuration of nucleons. The process is probabilistic, and the average chance of a reaction (the microscopic cross section) may be measured as a function of the projectile, projectile energy and target, either experimentally or by optical model potential calculations. The reaction rate is calculated from the microscopic cross section using the following equation:

$$R = \frac{J}{e} n_t \sigma \cdot 10^{28} \delta t \quad (1.1)$$

- R Reaction Rate (reactions per second)
- J Beam current (A)
- n_t Number density of target (atoms per cubic metre)
- σ Microscopic reaction cross section (barns)
- e Elementary charge (1.602177E-19C)
- δT Target thickness (m)

1.2 Decay and Activity Equations

1.2.1 Radioactive Decay

Radioactive decay is the random change in nucleons or energy state of an unstable nucleus. It is impossible to predict when a nucleus will decay, but the decay of a collection of nuclei is statistical in nature. The radioactivity and number of unstable nuclei at time t can be predicted using the decay constant λ for the radioactive isotope. This constant is defined as follows:

$$\lambda = -\frac{N'(t)}{N(t)} \quad (1.2)$$

The number of radioactive nuclei $N(t)$ at time t is given by the following equation, where $N(0)$ is the starting number of nuclei:

$$N(t) = N(0) \exp(-t\lambda) \quad (1.3)$$

The activity $A(t)$ of the radioactive nuclei is predicted at time t by using the following equations, where $N'(t)$ is the change in amount of nuclei with respect to time:

$$A(t) = -N'(t) = \lambda N(t) \quad (1.4)$$

$$A(t) = \lambda N(0) \exp(-t\lambda) \quad (1.5)$$

1.2.2 Bateman Equation for Radioactive Decay




Figure 1.2: An example decay chain from an unstable parent isotope, through unstable daughter isotopes ending with a stable daughter isotope.

The English mathematician Harry Bateman derived an equation to calculate the amount of each isotope in a decay chain, illustrated in Figure 1.2, at time t .

$$N_n(t) = \sum_{i=1}^{i=n} \left(\left(\prod_{j=i}^{j=n-1} \lambda_{(ij+1)} \right) \sum_{j=i}^{j=n} \left(\frac{N_{i0} \exp(-\lambda_j t)}{\prod_{p=i, p \neq j}^{p=n} (\lambda_p - \lambda_j)} \right) \right) \quad (1.6)$$

When a radioactive isotope decays, there may be more than one mode of decay, and this leads to branching factors. Pb-214 only decays via beta decay to Bi-214, giving a branching factor of 1.0, whereas Bi-214 has a 99.979% chance of decaying to Po-214 by beta decay and a 0.021% of emitting an alpha particle and decaying to Tl-210 (branching factors of 0.99979 and 0.00021 respectively) [5].

When a target material is irradiated, there is a source term for transmuted nuclei due to the irradiation. The daughter isotopes of these transmuted isotopes will also be affected by the irradiation and will transmute further, giving a source term for each daughter isotope as a result of the irradiation. Sources for each isotope in the decay chain, and branching factors between a parent isotope and its daughter isotope/s must be accounted for.

The Bateman equation was derived using Laplace transforms, and this same method has been used to develop a modified equation that incorporates branching factors and production rates for each isotope in the decay chain, as illustrated by Figure 1.3.




Figure 1.3: An example of several decay chains including branching factors and possible external source terms for each isotope on each chain.

1.2.3 Laplace Transform

Laplace Transforms are a useful mathematical tool, and allow ordinary differential equations to be solved by simple algebraic manipulation in the s domain. Bateman took advantage of Laplace Transforms in deriving his equation, and this is the method that has been taken here as well.

$$F(s) = \int_0^{\infty} f(t) \exp(-st) dt \quad (1.7)$$

1.2.4 Constructing the Differential Equations

The first step is to set up differential equations for the parent isotope, unstable daughter isotopes and stable daughter isotope. The parent isotope has a source term, due to production, and a loss term, due to decay. The unstable daughter isotopes have two source terms, from the production by irradiation induced transmutation

and the decay of preceding isotopes in the decay chain, and a loss term, due to decay. Finally, the stable daughter that finalizes the decay chain has two source terms (the same as the unstable daughters) but no loss term.

The variables (and vectors) used in these equations are defined as follows:

- $\vec{\lambda}$ vector containing isotope decay constants λ_i
- \vec{b} vector containing isotope to isotope branching factors b_i
- \vec{w} vector containing isotope production rates w_i
- t time at which activity/amount of isotope is measured
- $N_i(0)$ starting amount of the i th isotope
- $N_i(t)$ amount of the i th isotope at time t
- $N'_i(t)$ change in amount of the i th isotope, with respect to time, at time t

The differential equations for the parent isotope (first isotope), unstable daughter isotopes (i th isotopes) and stable, final, daughter isotope (z th isotope) in the time domain are as follows:

$$N'_1(t) = \omega_1 - \lambda_1 N_1(t) \quad (1.8)$$

$$N'_i(t) = \omega_i + b_{i-1} \lambda_{i-1} N_{i-1}(t) - \lambda_i N_i(t) \quad (1.9)$$

$$N'_z(t) = \omega_z + b_{z-1} \lambda_{z-1} N_{z-1}(t) \quad (1.10)$$

Applying the Laplace Transform to these three differential equations allows them to be manipulated and solved algebraically in the s -domain.

$$N_1(s) = \frac{1}{s + \lambda_1} N_1(0) + \frac{1}{s(s + \lambda_1)} \omega_1 \quad (1.11)$$

$$N_i(s) = \frac{1}{s(s + \lambda_i)} (\omega_i) + \frac{1}{s + \lambda_i} (b_{i-1} \lambda_{i-1} N_{i-1}(s)) + \frac{1}{s + \lambda_i} N_i(0) \quad (1.12)$$

$$N_z(s) = \frac{1}{s^2} \omega_z + \frac{1}{s} b_{z-1} \lambda_{z-1} N_{z-1}(s) + \frac{1}{s} N_z(0) \quad (1.13)$$

1.2.5 Numerical Inversion of the Laplace Transform

The Gaver-Stehfest[6] algorithm was developed in the 1960s and 1970s and is a method of calculating the inverse of a Laplace Transform in the real number domain. It is an easy to implement and reasonably accurate method, although it is an approximation to the real value. A comparison between an analytic and numeric inversion for the unstable isotope Po-218 is discussed at the end of this section.

$$f(t) \approx f_n(t) = \frac{\ln(2)}{t} \sum_{k=1}^{2n} a_k(n) F(s) \text{ where } n \geq 1, t > 0 \quad (1.14)$$

$$s = \frac{k \ln(2)}{t} \quad (1.15)$$

$$a_k(n) = \frac{(-1)^{(n+k)}}{n!} \sum_{j=\text{Floor}(\frac{k+1}{2})} j^{n+1} \binom{n}{j} \binom{2j}{j} \binom{j}{k-j} \quad (1.16)$$

The equation for the i th isotope may be calculated by recursively calculating the equations by numeric inversion, starting from the first (parent isotope) and inserting the result into each subsequent recursion until the i th isotope is reached (changing the equations appropriately for the parent, unstable daughter and stable daughter isotopes).

1.2.6 Analytic Solution by Partial Fraction Expansion

The equation for the i th isotope in the s domain can be written in full by substituting the preceding equation until the parent isotope is reached, and this full equation may be rearranged with the production amount of each isotope and starting amount of each isotope in individual terms. Each of these terms is multiplied by a fraction that can be expanded, using partial fractions, and inverted analytically.

This is illustrated with an example unstable isotope, fourth in the decay chain (including the parent isotope):

$$\begin{aligned} N_4(s) = & \frac{1}{(s + \lambda_1)(s + \lambda_2)(s + \lambda_3)(s + \lambda_4)} b_2 b_3 b_4 \lambda_1 \lambda_2 \lambda_3 N_1(0) \\ & + \frac{1}{(s + \lambda_2)(s + \lambda_3)(s + \lambda_4)} b_3 b_4 \lambda_2 \lambda_3 N_2(0) \\ & + \frac{1}{(s + \lambda_3)(s + \lambda_4)} b_4 \lambda_3 N_3(0) \\ & + \frac{1}{(s + \lambda_4)} N_4(0) \\ & + \frac{1}{s(s + \lambda_1)(s + \lambda_2)(s + \lambda_3)(s + \lambda_4)} b_2 b_3 b_4 \lambda_1 \lambda_2 \lambda_3 \omega_1 \\ & + \frac{1}{s(s + \lambda_2)(s + \lambda_3)(s + \lambda_4)} b_3 b_4 \lambda_2 \lambda_3 \omega_2 \\ & + \frac{1}{s(s + \lambda_3)(s + \lambda_4)} b_4 \lambda_3 \omega_3 \\ & + \frac{1}{s(s + \lambda_4)} \omega_4 \end{aligned} \quad (1.17)$$

An example stable isotope, fourth (last) in the decay chain (including the parent isotope):

$$\begin{aligned}
N_4(s) = & \frac{1}{s(s+\lambda_1)(s+\lambda_2)(s+\lambda_3)} b_2 b_3 b_4 \lambda_1 \lambda_2 \lambda_3 N_1(0) \\
& + \frac{1}{s(s+\lambda_2)(s+\lambda_3)} b_3 b_4 \lambda_2 \lambda_3 N_2(0) \\
& + \frac{1}{s(s+\lambda_3)} b_4 \lambda_3 N_3(0) \\
& + N_4(0) \\
& + \frac{1}{s^2(s+\lambda_1)(s+\lambda_2)(s+\lambda_3)} b_2 b_3 b_4 \lambda_1 \lambda_2 \lambda_3 \omega_1 \\
& + \frac{1}{s^2(s+\lambda_2)(s+\lambda_3)} b_3 b_4 \lambda_2 \lambda_3 \omega_2 \\
& + \frac{1}{s^2(s+\lambda_3)} b_4 \lambda_3 \omega_3 \\
& + \frac{1}{s^2} \omega_4
\end{aligned} \tag{1.18}$$

By using partial fraction expansion and standard Laplace Transforms, the set of equations below is used to calculate the amount of the m th isotope in the decay chain, providing the m th isotope is unstable.

$$N_m(t; \vec{\lambda}, \vec{b}, \vec{w}) = \sum_{k=1, m} r(k; \vec{\lambda}, \vec{b}) \left[f(t; k, m, \vec{\lambda}) N_k(0) + g(t; k, m, \vec{\lambda}) w_k \right] \tag{1.19}$$

$$r(k, m, \vec{\lambda}) = \begin{cases} \prod_{i=k, m-1} (b_{i+1} \lambda_i), & \text{if } k < m \\ 1, & \text{if } k = m \end{cases} \tag{1.20}$$

$$f(t; k, m, \vec{\lambda}) = (-1)^{m-k} \sum_{i=k, m} \left[\exp(-\lambda_i t) \prod_{j=k, m; j \neq i} \left(\frac{1}{\lambda_i - \lambda_j} \right) \right] \tag{1.21}$$

$$g(t; k, m, \vec{\lambda}) = \frac{1}{\prod_{i=k, m} \lambda_i} + (-1)^{m-k+1} \sum_{i=k, m} \left[\frac{1}{\lambda_i} \exp(-\lambda_i t) \prod_{j=k, m; j \neq i} \left(\frac{1}{\lambda_i - \lambda_j} \right) \right] \tag{1.22}$$

The set of equations below is used to calculate the amount of the m th isotope in the decay chain, where the m th isotope is stable.

$$N_m(t; \vec{\lambda}, \vec{b}, \vec{w}) = N_m + w_m t + \sum_{k=1, m-1} r(k; \vec{\lambda}, \vec{b}) \left[f(t; k, m-1, \vec{\lambda}) N_k(0) + g(t; k, m, \vec{\lambda}) w_k \right] \tag{1.23}$$

$$r(k, m, \vec{\lambda}) = \begin{cases} \prod_{i=k, m-1} (b_{i+1} \lambda_i), & \text{if } k < m \\ 1, & \text{if } k = m \end{cases} \tag{1.24}$$




Figure 1.4: Decay of Po-218: Analytic and Gaver-Stehfest Calculations [5]

$$f(t; k, m, \vec{\lambda}) = \frac{1}{\prod_{i=k,m} \lambda_i} + (-1)^{m-k+1} \sum_{i=k,m} \left[\frac{1}{\lambda_i} \exp(-\lambda_i t) \prod_{j=k,m; j \neq i} \left(\frac{1}{\lambda_i - \lambda_j} \right) \right] \quad (1.25)$$

$$g(t; k, m, \vec{\lambda}) = \frac{1}{\prod_{i=k,m} \lambda_i} t + \frac{\sum_{i=k,m} \left[\prod_{j=k,m; j \neq i} \lambda_j \right]}{\prod_{i=k,m} \lambda_i^2} + (-1)^{m-k+1} \sum_{i=k,m} \left[\frac{1}{\lambda_i^2} \exp(-\lambda_i t) \prod_{j=k,m; j \neq i} \left(\frac{1}{\lambda_i - \lambda_j} \right) \right] \quad (1.26)$$

1.2.7 Preference: Analytic over Numeric

The numeric solution only requires the equation to be solved in the s-domain; the Gaver-Stehfest algorithm performs the inversion. It is worth the extra effort to derive and implement an analytic solution, as the numeric is only an approximation. Examples of the pitfalls of the numeric solution are that it can give negative amounts of an isotope and the difference between the numeric and analytic calculated amounts can become quite large when the isotope decays away to a very small value. Figure 1.4 shows the predicted decay of a sample of Po-218 irradiated for 1,000s, and sampled until 10,000s. In the region between 4,000s and 9,000s the amount from the numeric calculation drops below zero, whereas the analytic calculation remains above zero, as would be expected.




Figure 1.5: Flow chart of major processes in the Activity code

1.3 Computational Methods

1.3.1 Activity Code

The Activity program has been developed in Fortran and takes advantage of MPI (Message Parsing Interface) to speed up calculation times by allowing the use of multiple processes in parallel. It has a self contained maths library, although this could be improved in the future by using optimised maths libraries for certain functions (e.g. matrix operations).

The code was developed on a Debian version of Linux, but it should be supported on other variants of Linux and Unix, and does not require any specialist hardware.

The user is required to prepare an input file that contains the instructions required to perform a calculation. In addition to the input file, the user must provide an EXYZ ion trajectory file output by SRIM. Activity will read in the user input file, and the SRIM and data files listed within, before performing the calculation. Figure 1.5 shows a flowchart of the major steps the code performs.

There are various settings in the user input file, but the main ones relating to the simulated experiment are:

- element composition of target (percentage by mass)
- beam flux (current), energy, duration and area on target
- activity measurement time (end of the “experiment”)
- material density
- target thickness

Several data files are generated by Activity and, if the user has matplotlib [7], charts will be created too. The most relevant to the user are:

- gammaLines.dat - tally into discrete bins of predicted gamma counts

- ionTraj.dat - the averaged ion trajectory used in the calculation
- isotopeActivityFileG.dat - a large data file detailing the activity of every predicted radioactive isotope in the target at user specified times following irradiation

The charts include:

- activityTop5.png - activity of the top 5 active isotopes as a function of time after irradiation starts
- gammaLines.png - predicted gamma spectrum expected at the “experiment end time”

The Activity code uses the equations derived above to calculate the amount and activity of each isotope in the calculation. One problem with the original Bateman equation that also exists in the set of modified Bateman equations is that two different isotopes with the same decay constant will cause a singularity and a halt in the calculation. The activity code loops through all the decay constants in use before it attempts to run the calculation. If any isotope decay constants match they are varied by a small amount relative to the decay constant. It repeats this process until all decay constants are unique before proceeding.

1.3.2 Approximations

The accuracy of the Activity code is dependant on the input files provided by the user and the method used to calculate the reaction rates and resulting activity. The TENDL proton database consists of experimental measured cross sections as well as values calculated using the optical model potential. Using the latest database is recommended.

SRIM uses the binary collision approximation to simulate ion transport. It is a well tested code that has been used for many years. One limitation is that the structure of the material is not taken into account. This would have an impact on a user of the Activity program if they were trying to calculate, for example, whether a FCC (face centered cubic) steel would be irradiated differently when compared to a BCC (body centered cubic) steel. The Activity code would determine the activity of the steel as a function of the ion current, ion type and the density, thickness and composition of the steel, not its structure.

This version of the Activity code averages the path of all the SRIM simulated ions, rather than treating each ion differently. This may or may not have an impact on the results. If a new version of the code is developed there would be an option to calculate reaction rates for each individual simulated ion, and a comparison could then be made to the calculations using the averaged path of a set of ions.

The final approximation would be to use the numeric solution to the activity equations, although the analytic solution is forced within the code unless it returns a failed result.

1.3.3 Results

A target of high purity Iron was irradiated with 36 MeV protons by the University of Birmingham Scanditronix MC-40 Cyclotron. The target was 0.5mm thick and was irradiated at a current of 0.5 micro Amps for 300 seconds, irradiating approximately 0.25g of Iron. A high purity Germanium detector was used to measure the gamma peaks three days after irradiation.

The peak that dominated the readings was the 931 keV Cobalt 55 line. After calibrating the detector and adjusting the readings, this peak was measured at 44,300Bq+/-2,000Bq. The activity of this peak as predicted by the Activity code was 44,565Bq.

A target of high purity Molybdenum was irradiated with 13 MeV protons by the University of Birmingham Scanditronix MC-40 Cyclotron. The target was 0.5mm thick and was irradiated at a current of 5 micro Amps

Table 1.1: Gamma peaks predicted and measured for a 13 MeV ion irradiated sample of Mo

Gamma Energy (keV)	Predicted Activity (Bq)	Experimental Activity (Bq)
766	4.45E6	5.11E5+/-2.5E4
778	6.14E6	1.36E6+/-6.8E4
812	5.04E6	1.15E6+/-5.8E4
850	6.00E6	1.39E6+/-7.0E4
126	9.33E5	2.10E5+/-1.1E4




Figure 1.6: Sample Activity code output chart for the top five most active isotopes for Iron irradiated by 36MeV protons.

for 1500 seconds, irradiating approximately 0.3g of Molybdenum. A high purity Germanium detector was used to measure the gamma peaks three days after irradiation.

Five peaks were of interest, and these are listed in Table 1.1.




Figure 1.7: Sample Activity code output chart for the expected gamma lines to be measured for Iron irradiated by 36MeV protons.




Figure 1.8: Sample Activity code output chart for the top five most active isotopes for Molybdenum irradiated by 13MeV protons.




Figure 1.9: Sample Activity code output chart for the expected gamma lines to be measured for Molybdenum irradiated by 13MeV protons.

Chapter 2

Installation and Using the Activity Code

2.1 Getting Started

2.1.1 Prerequisites

This code was designed to run on Linux and has been developed and tested on the Debian based distribution Mint. It has not been tested on any other versions of Linux or Unix, nor has it been tested on cygwin.

The minimal requirements of the code are:

- 250MB per process + 200MB overhead
- 1GB Hard Drive space
- Ideally a modern multicore processor
- Fortran90
- OpenMPI

Optional:

- GNUPlot
- Python & Matplotlib

2.2 Installing Activity

2.2.1 Download Source Code

The most recent source code is available for download from github. It may be downloaded and extracted using the terminal and the commands in listing 1.

Listing 2.1: bash

```

1 cd ~
2 mkdir -p ~/activity/tar
3 cd ~/activity/tar
4 wget https://github.com/BenPalmer1983/activity/raw/master/activity.tar.gz
5 tar -xzf activity.tar.gz

```

2.2.2 Compile Source Code

Once the source files have been downloaded and extracted, run the install script then remove the downloaded files (if you need to free up space). The terminal commands are given in listing 2. Providing the prerequisites are available, the process will only take a few minutes (with a reasonable Internet connection).

Listing 2.2: bash

```

1 cd ~/activity/tar
2 ./install.sh #choose installation directory when prompted; default is ~/activity
3 rm -R ~/activity/tar

```

The activity code is now installed and the activity.x binary may be executed on the terminal. If the terminal fails to find the binary, the .bashrc or .bash_profile file must be edited, or a symlink created from the bin file to a valid and accessible bin directory.

2.3 Input File

The input file sets the target starting composition, the beam and simulated experiment settings as well as the paths to the required data files.

#elements

The elements keyword instructs the code to read in the composition of the target material. The elements and their percentage by weight are listed under the keyword.

Listing 2.3: Input File

```

1 #elements
2 Fe 72
3 Cr 18
4 Ni 8
5 Mg 2

```

#isotopes #decaymodes #gammaenergies #xsfiles

Four sets of data files are distributed with the code, and the paths to these files are defined underneath each keyword. N.B. the path for the xsfiles is to the directory that holds all the cross section data files, rather than each individual file.

Listing 2.4: Input File

```

1 #isotopes
2 "/home/ben/activity/data/isotopes.txt"
3 #decaymodes
4 "/home/ben/activity/data/decaymodes.txt"

```

```
5 #gammaenergies
6 "/home/ben/activity/data/gammaenergies.txt"
7 #xsfiles
8 "/home/ben/activity/data/xs"
```

#trajfile

The SRIM exyz file used for the calculation is pointed to under this keyword.

Listing 2.5: Input File

```
1 #trajfile
2 "/home/ben/activity/examples/Fe36MeV.exyz"
```

#polyfitorder *#integrationgranularity*

The code calculates the reaction rate of each projectile and target nucleus by first fitting a polynomial to the energy/depth data file. This polynomial gives an average path for the projectiles travelling through the target, allowing $E(x)$ to be calculated quickly. The order of the polynomial is selected here, and a 5th order should give a reasonable fit. The integration granularity keyword determines how many sections the polynomial is split up into. The energy, cross section for each section and the beam settings are used to calculate the reaction rate for each section, and these are all summed up to give the overall reaction rate.

Listing 2.6: Input File

```
1 #polyfitorder
2 5
3 #integrationgranularity
4 10
```

#beamflux *#beamenergy* *#beamduration* *#beamarea*

These keywords are self explanatory and control the beam settings used in the calculation.

Listing 2.7: Input File

```
1 #beamflux
2 0.5 uA
3 #beamenergy
4 36 MeV
5 #beamduration
6 300 s
7 #beamarea
8 100 mm2
```

#amtime *#timestep*

The time at which the activity of the sample is measured is set by the *#amtime* keyword. The Activity code calculates the activity of each radioactive isotope in the calculation from the time the beam starts until the activity measured time; and the intervals at which these calculations are made is determined by the *#timestep* keyword.

Listing 2.8: Input File

```
1 #amtime
2 260000 s
```

```
3 #timestep
4 1000 s
```

#projectile

The current version only supports protons as the projectile, therefore the only possible atomic and mass number combination for a projectile is 1 1.

Listing 2.9: Input File

```
1 #projectile
2 1 1
```

#targetthickness #materialdensity

Both the target thickness and density of the material it is made from are input with these keywords.

Listing 2.10: Input File

```
1 #targetthickness
2 0.5 mm
3 #materialdensity
4 8000 kgm3
```

#vpi

This keyword is only being used for a feature currently under testing, so the keyword can be omitted from the input file.

Listing 2.11: Input File

```
1 #vpi
2 60.2
```

#individualisotopeactivity #verboseterminal

If the *#individualisotopeactivity* keyword is followed by *yes*; additional data files are output containing the activity of each individual isotope in the calculation. *#verboseterminal* followed by *yes* will output more verbosely to the terminal.

Listing 2.12: Input File

```
1 #individualisotopeactivity
2 yes
3 #verboseterminal
4 yes
```

#targetdpa

This keyword is only being used for a feature currently under testing, so the keyword can be omitted from the input file.

Listing 2.13: Input File

```
1 #targetdpa
2 0.0
```

```
#gammachartresolution
```

The gamma chart is created by tallying the gamma values into bins, and this figure specifies the resolution(number of bins) used to create the output chart.

Listing 2.14: Input File

```
1 #gammachartresolution  
2 200
```

2.4 Acknowledgements

We would like to thank and acknowledge the input and advice of the following:

- Dr Chris Cooper and John Hewett for irradiation activation data points.
- The University of Birmingham for providing the funding for this project.

Appendices

Appendix A

Example Input File

A.1 Iron 36MeV Proton Beam

Listing A.1: Fe36MeV.in

```
1  #elements
2  Fe 100
3  #isotopes
4  "/home/ben/activity/data/isotopes.txt"
5  #decaymodes
6  "/home/ben/activity/data/decaymodes.txt"
7  #gammaenergies
8  "/home/ben/activity/data/gammaenergies.txt"
9  #xsfiles
10 "/home/ben/activity/data/xs"
11 #trajfile
12 "/home/ben/activity/examples/Fe36MeV.exyz"
13 #polyfitorder
14 5
15 #integrationgranularity
16 10
17 #beamflux
18 0.5 uA
19 #beamenergy
20 36 MeV
21 #beamdURATION
22 300 s
23 #beamarea
24 100 mm2
25 #amtime
26 260000 s
27 #timestep
28 1000 s
29 #projectile
30 1 1
31 #targetthickness
32 0.5 mm
33 #materialdensity
34 8000 kgm3
```



```
35 #upi
36 60.2
37 #individualisotopeactivity
38 yes
39 #verboseterminal
40 yes
41 #targetdpa
42 0.0
43 #gammachartresolution
44 200
```

Appendix B

Fortran 90 Code

B.1 Fortran 90 Implementation of Analytic Method

Listing B.1: Fortran 90

```
1  Type :: decayChainObj
2      Real(kind=DoubleReal) :: time = 0.0D0
3      !Real(kind=DoubleReal) :: productionRate = 0.0D0
4      Integer(kind=StandardInteger) :: isotopes
5      Character(Len=16), Dimension(1:100) :: label
6      Real(kind=DoubleReal), Dimension(1:100) :: productionRate = 0.0D0
7      Real(kind=DoubleReal), Dimension(1:100) :: branchFactor = 1.0D0 ! from isotope parent
8      Real(kind=DoubleReal), Dimension(1:100) :: decayConstant = -1.0D0 ! negative for stable
9      Real(kind=DoubleReal), Dimension(1:100) :: halfLife = -1.0D0 ! negative for stable
10     Real(kind=DoubleReal), Dimension(1:100) :: amountStart = 0.0D0
11     Real(kind=DoubleReal), Dimension(1:100) :: amountEnd = 0.0D0
12 End Type
13
14 Subroutine CalcIsotopeChain(decayChain)
15 ! Uses inverse laplace transform to calculate isotope amounts at time t (after time = 0)
16 ! t time in seconds after t=0
17 ! w production rate of parent isotope
18 ! isotope chain data
19     Implicit None ! Force declaration of all variables
20 ! Vars In/Out
21     Type(decayChainObj) :: decayChain
22 ! Vars Private
23     Integer(kind=StandardInteger) :: i
24     Real(kind=DoubleReal) :: t, nEnd
25     Real(kind=DoubleReal), Dimension(1:100) :: W ! Production Rate
26     Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
27     Real(kind=DoubleReal), Dimension(1:100) :: N ! Starting number of atoms
28     Real(kind=DoubleReal), Dimension(1:100) :: B ! Exp
29 ! Complete decay chain data
30     Call decayChainComplete(decayChain)
31 ! store input in shortned name arrays to make equations clearer
32     t = decayChain%time
33     Do i=1,decayChain%isotopes
34         W(i) = decayChain%productionRate(i)
```

```

35     L(i) = decayChain%decayConstant(i)
36     B(i) = decayChain%branchFactor(i)
37     N(i) = decayChain%amountStart(i)
38 End Do
39 ! Break infinities
40 Call DecayBreakInfinities(L,decayChain%isotopes)
41 ! Run analytic calculations
42 ! Loop through isotopes
43 ! Using  $L^{-1}(1/(q+ps)) = (1/p) \cdot \exp(-1 \cdot (q/p) \cdot t)$  and partial fractions
44 Do i=1,decayChain%isotopes
45     nEnd = CalcIsotopeChainCalc(t,i,W,L,N,B)
46     decayChain%amountEnd(i) = nEnd
47 End Do
48 End Subroutine CalcIsotopeChain
49
50 Subroutine decayChainComplete(decayChain)
51 ! Completes the decay chain object
52 Implicit None ! Force declaration of all variables
53 ! Vars In/Out
54 Type(decayChainObj) :: decayChain
55 ! Vars Private
56 Integer(kind=StandardInteger) :: i, n
57 n = 0
58 Do i=1,100
59     n = n + 1
60     If(decayChain%decayConstant(i).eq.-1.0D0.and.decayChain%halfLife(i).gt.0.0D0)Then
61 ! complete decay constant from half life
62         decayChain%decayConstant(i) = lnTwo/decayChain%halfLife(i)
63     End If
64     If(decayChain%halfLife(i).le.0.0D0.and.decayChain%decayConstant(i).gt.0.0D0)Then
65 ! complete decay constant from half life
66         decayChain%halfLife(i) = lnTwo/decayChain%decayConstant(i)
67     End If
68 ! Adjust for stable isotope
69     If(decayChain%decayConstant(i).lt.0.0D0)Then
70         decayChain%halfLife(i) = -1.0D0
71         decayChain%decayConstant(i) = 0.0D0
72     End If
73     If(decayChain%halfLife(i).lt.0.0D0)Then
74         decayChain%halfLife(i) = -1.0D0
75         decayChain%decayConstant(i) = 0.0D0
76     End If
77 ! Break out if stable
78     If(decayChain%decayConstant(i).eq.0.0D0)Then
79         Exit
80     End If
81 End Do
82 decayChain%isotopes = n
83 End Subroutine decayChainComplete
84
85 Subroutine DecayBreakInfinities(L,n)
86 !
87 Implicit None ! Force declaration of all variables

```

```

88 ! Vars In/Out
89   Real(kind=DoubleReal), Dimension(:) :: L ! Lambda
90   Integer(kind=StandardInteger) :: n
91 ! Vars Private
92   Integer(kind=StandardInteger) :: i,j
93   Logical :: breaking
94 ! Loop and alter matching decay constants slightly
95   breaking = .true.
96   Do While(breaking)
97     breaking = .false.
98     Do i=1,n-1
99       Do j=i+1,n
100        If(L(i).eq.L(j))Then
101          breaking = .true.
102          L(i) = L(i)*1.0000001D0 ! Vary by 0.00001%
103        End If
104      End Do
105    End Do
106  End Do
107  End Subroutine DecayBreakInfinites
108
109  Function CalcIsotopeChainCalc(t,m,W,L,N,B) Result (nEnd)
110    Implicit None ! Force declaration of all variables
111  ! Vars In
112    Real(kind=DoubleReal) :: t
113    Integer(kind=StandardInteger) :: m
114    Real(kind=DoubleReal), Dimension(1:100) :: W ! Production Rate
115    Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
116    Real(kind=DoubleReal), Dimension(1:100) :: N ! Starting number of atoms
117    Real(kind=DoubleReal), Dimension(1:100) :: B ! Branch factor
118  ! Vars Out
119    Real(kind=DoubleReal) :: nEnd
120  ! Vars Private
121    Integer(kind=StandardInteger) :: i, j, k, z
122    Real(kind=DoubleReal) :: mult, multR
123    Real(kind=DoubleReal) :: nChange
124  ! Init
125    nEnd = 0.0D0
126  ! -----
127  ! UNSTABLE Isotopes
128  ! -----
129    If(L(m).gt.0.0D0)Then
130  ! Loop through terms
131      Do k=1,m
132        multR = CalcIsotopeChainMultR(k,m,L,B)
133        mult = multR * N(k)
134  ! decay of starting matter
135        nChange = CalcIsotopeChainF_Unstable(k,m,t,mult,L)
136        nEnd = nEnd + nChange
137  ! production term
138        mult = multR * W(k)
139        nChange = CalcIsotopeChainG_Unstable(k,m,t,mult,L)
140        nEnd = nEnd + nChange

```

```

141         print *,k,nEnd
142     End Do
143 Else
144 ! -----
145 ! STABLE Isotopes
146 ! -----
147 ! Loop through terms
148     nEnd = N(m)+t*W(m)
149     Do k=1,m-1
150         multR = CalcIsotopeChainMultR(k,m,L,B)
151         mult = multR * N(k)
152 ! decay of starting matter
153         nChange = CalcIsotopeChainF_Stable(k,m,t,mult,L)
154         nEnd = nEnd + nChange
155 ! production term
156         mult = multR * W(k)
157         nChange = CalcIsotopeChainG_Stable(k,m,t,mult,L)
158         nEnd = nEnd + nChange
159         print *,k,nEnd
160     End Do
161 End If
162 End Function CalcIsotopeChainCalc
163
164 Function CalcIsotopeChainMultR(k,m,L,B) Result (multR)
165 ! Vars In
166     Integer(kind=StandardInteger) :: k, m
167     Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
168     Real(kind=DoubleReal), Dimension(1:100) :: B ! Branching Factor
169 ! Vars Out
170     Real(kind=DoubleReal) :: multR
171 ! Private
172     Integer(kind=StandardInteger) :: i
173 ! Result
174     multR = 1.0D0
175     If(k.lt.m)Then
176         Do i=k,m-1
177             multR = multR * B(i+1) * L(i)
178         End Do
179     End If
180 End Function CalcIsotopeChainMultR
181
182 ! -----
183 ! Unstable Isotope Functions
184 ! -----
185
186 Function CalcIsotopeChainF_Unstable(k,m,t,mult,L) Result (nChange)
187 ! Vars In
188     Integer(kind=StandardInteger) :: k, m
189     Real(kind=DoubleReal) :: t, mult
190     Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
191 ! Vars Out
192     Real(kind=DoubleReal) :: nChange
193 ! Private

```

```

194     Real(kind=DoubleReal) :: multP
195     Real(kind=DoubleReal) :: p, q, r, s
196     Integer(kind=StandardInteger) :: i, j
197 ! Calculate isotope amount change
198     nChange = 0.0D0
199     multP = (-1.0D0)**(m-k)
200 ! Loop through pfrac
201     Do i=k,m
202         r = 1.0D0
203         Do j=k,m
204             If(j.ne.i)Then
205                 r = r * (L(i)-L(j))
206             End If
207         End Do
208         nChange = nChange + (1.0D0/r)*exp(-1.0D0*L(i)*t)*multP*mult
209     End Do
210 End Function CalcIsotopeChainF_Unstable
211
212 Function CalcIsotopeChainG_Unstable(k,m,t,mult,L) Result (nChange)
213 ! Vars In
214     Integer(kind=StandardInteger) :: k, m
215     Real(kind=DoubleReal) :: t, mult
216     Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
217 ! Vars Out
218     Real(kind=DoubleReal) :: nChange
219 ! Private
220     Real(kind=DoubleReal) :: multP
221     Real(kind=DoubleReal) :: p, q, r, s
222     Integer(kind=StandardInteger) :: i, j
223 ! Calculate isotope amount change
224     nChange = 0.0D0
225     multP = (-1.0D0)**(m-k+1)
226 ! term A
227     r = 1.0D0
228     Do i=k,m
229         r = r * L(i)
230     End Do
231     nChange = nChange + (1.0D0/r)*mult
232 ! Loop through pfrac
233     Do i=k,m
234         r = 1.0D0*L(i)
235         Do j=k,m
236             If(j.ne.i)Then
237                 r = r * (L(i)-L(j))
238             End If
239         End Do
240         nChange = nChange + (1.0D0/r)*exp(-1.0D0*L(i)*t)*multP*mult
241     End Do
242 End Function CalcIsotopeChainG_Unstable
243
244 ! -----
245 ! Stable Isotope Functions
246 ! -----

```

```

247
248 Function CalcIsotopeChainF_Stable(k,mIn,t,mult,L) Result (nChange)
249 ! Vars In
250 Integer(kind=StandardInteger) :: k, mIn
251 Real(kind=DoubleReal) :: t, mult
252 Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
253 ! Vars Out
254 Real(kind=DoubleReal) :: nChange
255 ! Private
256 Integer(kind=StandardInteger) :: m
257 Real(kind=DoubleReal) :: multP
258 Real(kind=DoubleReal) :: p, q, r, s
259 Integer(kind=StandardInteger) :: i, j
260 ! In
261 m = mIn-1
262 ! Calculate isotope amount change
263 nChange = 0.0D0
264 multP = (-1.0D0)**(m-k+1)
265 ! term A
266 r = 1.0D0
267 Do i=k,m
268 r = r * L(i)
269 End Do
270 nChange = nChange + (1.0D0/r)*mult
271 ! Loop through pfrac
272 Do i=k,m
273 r = 1.0D0*L(i)
274 Do j=k,m
275 If(j.ne.i)Then
276 r = r * (L(i)-L(j))
277 End If
278 End Do
279 nChange = nChange + (1.0D0/r)*exp(-1.0D0*L(i)*t)*multP*mult
280 End Do
281 End Function CalcIsotopeChainF_Stable
282
283
284 Function CalcIsotopeChainG_Stable(k,mIn,t,mult,L) Result (nChange)
285 ! Vars In
286 Integer(kind=StandardInteger) :: k, mIn
287 Real(kind=DoubleReal) :: t, mult
288 Real(kind=DoubleReal), Dimension(1:100) :: L ! Lambda
289 ! Vars Out
290 Real(kind=DoubleReal) :: nChange
291 ! Private
292 Integer(kind=StandardInteger) :: m
293 Real(kind=DoubleReal) :: multP
294 Real(kind=DoubleReal) :: p, q, r, s
295 Integer(kind=StandardInteger) :: i, j
296 ! In
297 m = mIn-1
298 ! Calculate isotope amount change
299 nChange = 0.0D0

```

```

300     multP = (-1.0D0)**(m-k)
301 ! term A
302     r = 1.0D0
303     Do i=k,m
304         r = r * L(i)
305     End Do
306     nChange = nChange + (1.0D0/r)*t*mult
307 ! term B
308     p = CalcIsotopeChainC(L,k,m)
309     q = 1.0D0
310     Do i=k,m
311         q = q*L(i)*L(i)
312     End Do
313     r = (-1.0D0)*(p/q)
314     nChange = nChange + r*mult
315 ! Loop through pfrac
316     Do i=k,m
317         r = 1.0D0*L(i)*L(i)
318         Do j=k,m
319             If(j.ne.i)Then
320                 r = r * (L(i)-L(j))
321             End If
322         End Do
323         nChange = nChange + (1.0D0/r)*exp(-1.0D0*L(i)*t)*multP*mult
324     End Do
325 End Function CalcIsotopeChainG_Stable
326
327 Function CalcIsotopeChainC(L,k,m) Result (numerator)
328 ! Calculates numerator in isotope activity function
329 Implicit None ! Force declaration of all variables
330 ! Vars In
331     Real(kind=DoubleReal), Dimension(:) :: L
332     Integer(kind=StandardInteger) :: k, m
333 ! Vars Out
334     Real(kind=DoubleReal) :: numerator
335 ! Vars Private
336     Integer(kind=StandardInteger) :: i, j
337     Real(kind=DoubleReal) :: tempMult
338     numerator = 0.0D0
339     Do i=k,m
340         tempMult = 1.0D0
341         Do j=k,m
342             If(j.ne.i)Then
343                 tempMult = tempMult * L(j)
344             End If
345         End Do
346         numerator = numerator + tempMult
347     End Do
348 End Function CalcIsotopeChainC

```

Bibliography

- [1] A. J. Koning D. Rochman. “Modern Nuclear Data Evaluation With The TALYS Code System”. In: *Nuclear Data Sheets* 113 (2012).
- [2] J. P. Biersack James F. Ziegler M. D. Ziegler. “SRIM - The stopping and range of ions in matter”. In: *Nuclear Instruments and Methods in Physics Research Section B* 268 (2010), pp. 1818–1823.
- [3] A. A. Sonzogni. *Nuclear Data Services*. *www-nds.iaea.org*. 2006. URL: <http://www-nds.iaea.org/ndspub/download-endf/ENDF-B-VII.0/decay-index.htm> (visited on 08/14/2015).
- [4] M. Herman M. B. Chadwick. “ENDF/B-VII.0: Next generation evaluated nuclear data library for nuclear science and technology.” In: *Nucl. Data Sheets* 107 (2006), p. 2931.
- [5] P. Blaise M. Coste A. Courcelle T.D. Huynh C. Jouanne P. Leconte O. Litaize S. Mengelle G. Nogure J-M. Ruggiri O. Srot J. Tommasi C. Vaglio J-F. Vidal A. Santamarina D. Bernard. *The JEFF-3.1.1 Nuclear Data Library*. 2009. ISBN: 978-92-64-99074-6.
- [6] H. Stehfest. “Algorithm 368: Numerical Inversion of Laplace Transform”. In: *Communications of the ACM* 13 (1970), pp. 47–49.
- [7] J. D. Hunter. “Matplotlib: A 2D graphics environment”. In: *Computing In Science & Engineering* 9 (2007), pp. 90–95.